
;----------------- Sintetizador por Randomizacao de harmonicos ------- by: Sergim Veiga 
<CsoundSynthesizer>
<CsOptions> 
-d            ;  \
-odac          ;  \
</CsOptions>    ;   <---- Cabeçalho com saída DAC 48kHz, stereo e taxa de cotrole 1 por 1.
<CsInstruments>;  /
sr = 48000   ;  /
ksmps = 1
nchnls = 2

FLcolor  255,   255,   255 ; <---- determina cor branca para a janela
FLpanel "Sintetizador por Randomizacao de harmonicos",810,400,0,0; <-- Abre um painel

idvol  FLvalue   "0-5000",     70,       30,    725,  20
idgama FLvalue   "1-512",     70,       30,    725,  80  ;  <--- criam monitoramento
idfrgr  FLvalue   "1-200Hz",     70,       30,    725,  140;        de valores
idfrfu  FLvalue   "1-12000Hz",     70,       30,    725,  200
idT60   FLvalue   ".01-20s",     70,       30,    725,  260

gk1,ih1 FLslider "Volume", 1,5000,-1,25, idvol, 700,30, 20,20
gk2,ih2 FLslider "Gama de harmonicos", 1,512,-1,25, idgama, 700,30, 20,80
gk3,ih3 FLslider "Frequencia das randomizacoes", 1,200,-1,25, idfrgr, 700,30, 20,140
gk4,ih4 FLslider "Frequencia fundamental", 1,12000,-1,25, idfrfu, 700,30, 20,200
gk5,ih5 FLslider "Tempo de reverberacao", .01,20,-1,25, idT60, 700,30, 20,260

FLsetVal_i   2000, ih1
FLsetVal_i   12, ih2
FLsetVal_i   8, ih3   ;  <------ determinam valores iniciais (default)
FLsetVal_i   440, ih4
FLsetVal_i   3, ih5

ihShow FLbox "Por: Sergio de Alencastro Veiga Filho",7,10,20,500,30,150,340

FLrun  ; <----- Fecha o painel e determina abrí-lo.

ga1 init  0   ;   <----  Variável global para o sinal de Reverberação

instr 1
kamp = gk1

k1 randomh 1, gk2, gk3   ;   <----  Randomizador

kcps = gk4*int(k1) ;  <--- transformação em números inteiros que multiplicam a frequência

a1 oscili kamp, kcps, 1  ;   <----  Oscilador

a2 linseg 0,.004,1,p3-.008,1,.004,0
outs a1*a2,a1*a2

ga1 = ga1+(a1*a2)
endin
instr 99 ; <-- instrumento para a reverberação

a1 reverb ga1,gk5 ;   <---- gk5 determina o T60 (tempo de atenuação em 60dB)
outs a1*.2,a1*.2

ga1 = 0
endin
</CsInstruments>
<CsScore>
f1 0 16384 10 1   ;  <----  Função senoidal
i99 0 3600    ;   <---- duração de 3600 segundos (1 hora) para a reverberação
i1 0 3600     ;   <---- duração de 3600 segundos (1 hora) para o instrumento
e
</CsScore>
</CsoundSynthesizer>

